
BACKGROUND INFORMATION ON
THE DEVELOPMENT OF THE NSSQ

The NSSQ was developed in 1980 to measure multiple
components of social support in a format that allows
respondents to list and rate their own social support
network members on functional properties of social
support (e.g., emotional and tangible support) and
according to some network properties (e.g., stability of
relationships, frequency of contact).  In addition,
questions about recent losses of supportive relationships
provide descriptive data.

The conceptual basis for the NSSQ, a description of the
instrument, and the results of the first phase of testing the
reliability and validity of the instrument were presented in
Nursing Research (Norbeck, Lindsey, & Carrieri, 1981).
Minor modifications were made in the instrument in 1982,
and a second phase of testing further examined validity
and provided normative data from employed adults
(Norbeck, Lindsey, & Carrieri, 1983). 

The 1995 Format Revision

In 1995 the NSSQ was revised to make the format of the
instrument, scoring sheet, and scoring instructions
compatible with the Window’s version of SPSS for data
entry and analysis.  The substantive content of the
instrument was not changed.

There are three differences between the 1980/82 and
1995 versions of the NSSQ.  First, the 5-point rating scale
for Questions 1-6 was changed from a 1-5 scale to a 0-4
scale in the 1995 version of the instrument itself to
eliminate the need for the scoring adjustment that had
been introduced in 1984.  Second, the printed format of
the instrument and scoring sheet was  changed to use
variable names rather than column numbers for data
entry.  Third, two of the original subscales, Affect and
Affirmation, were combined into a single subscale,
Emotional Support, in the Scoring Instructions.  Empirical
work with the instrument had shown that Affect and
Affirmation were too highly intercorrelated to warrant
keeping them as separate variables, and  the results of
factor analysis (see Appendix A) confirmed a two factor
solution of Emotional Support and Tangible Support

 
(Aid).  These 1995 scoring instructions should be used
with the 1995 version of the NSSQ.  For data collected
with the 1980/82 version of the NSSQ, the additional
instructions in Appendix B must be carried out prior to
applying the scoring  instructions presented here.  

Permission to use the NSSQ. Permission is hereby
granted to researchers and clinicians to copy and use the
1995 version of the NSSQ provided that the 1995 scoring
instructions are used.

SCORING THE NSSQ

The NSSQ can be scored directly from the questionnaire
or the responses may be transferred to a one-page
scoring sheet (see Figure 1).  The advantages of using
the scoring sheet are that the data are fully represented
on a single page and additional variables related to
source of support can be calculated.

SPSS program statements were developed for the NSSQ
to calculate variables, subscales, and source-specific
scores.  In addition, these statements check for the types
of errors in the data that result in scores outside the
possible range for calculated scores.  The SPSS program
statements for the NSSQ are provided in Appendix C. 

Direct Scoring Method

Table 1 presents a Computer Code Book for the data
obtained from the NSSQ.  The items are listed in the
order that they appear on the questionnaire.

"Number Listed in Network" [NOLISTED] is the number
of entries made on the network list.  Scores for this item
range from 1-24, however, a few subjects list more than
24 persons.  In this case, the actual number listed is
entered, and the ratings for these additional persons are
calculated into the scores for Questions 1-8.

For Questions 1-8, the scores are obtained, first,  by
adding the ratings for each question on pages 2-5 and
entering the sums in the columns indicated [EMO1,
EMO2, etc. to FREQCON].

"Source Category" [SOU1 to SOU24] is obtained by
entering the appropriate category code for each of the
first 24 network members listed.  In the Relationship
column of the personal network list on the NSSQ,
subjects have indicated the type of relationship.  These
entries are coded as follows:

0 = none (or a study-specific category)
1 = spouse or partner
2 = family or relatives
3 = friends
4 = work or school associates
5 = neighbors
6 = health care providers
7 = counselor or therapist
8 = minister, priest or rabbi
9 = other (e.g., God, pets)

The first of the recent loss items on page 6 is coded 0 =
no, 1 = yes [LOSS].  The total number of persons
indicated in Question 9a is entered for "Number Lost"
[LOSSNO].  The number indicated for each category is
also entered [LOSS1 to LOSS9].  The number checked
by the subject for Question 9b is entered for "Amount of
Support Lost" [LOSSAMT].

For subjects who answer Question 9 with a "no"
response, special scoring for Questions 9a and 9b must
be done to avoid subsequent problems with missing data
on the Total Loss variable.  Since these subjects do not
answer Questions 9a and 9b, answers indicating no
losses are supplied.  For each category of 9a, a "0"
response is entered [LOSS1 to LOSS9], and for "Number
Lost [LOSSNO], a "0" is entered.  For Question 9b, a "0"
is entered for "amount of Support Lost" [LOSSAMT].

Scoring Sheet Method

Prior to transferring the subject's responses to the scoring
sheet, the Source Categories are coded (0-9) as
indicated above.  These code numbers are entered in the
first column of the scoring sheet entitled "Relationship."

"Subject Number" [IDNO] and "Number in Network"
[NOLISTED] are obtained from the Personal Network
page of the NSSQ.  The subject's ratings for Questions 1-
8 are transferred to the scoring sheet.  For those subjects
who listed more than 24 network members, an additional
scoring sheet is used, and the totals of both sheets are
calculated.  the columns for Questions 1-8 are added on
the scoring sheet and the sums entered in the "Question
Totals” row.

The coding and entering of data from the recent loss
items follows the same procedure described in the direct
scoring section.  These scores are entered along the
bottom of the scoring sheet [LOSS to LOSSAMT].

The right side of the scoring sheet comprises the
additional scoring that is not easily calculated in the direct
scoring method. Table 2 presents a Computer Code Book
for these data.  "Person Totals” are the sums across the

rows for Questions 1-6 for each person listed in the
network [PER1 to PER24].  Frequency of contact with
individual network members, “Individual Contact,” is taken
directly from the ratings for Question 8 [CON1 to CON24].
By using the information about source of support [SOU1
to SOU24], the computer can be directed to calculate
Person Totals and Individual Contact for each Source
Category (e.g., the amount of functional support from
friends vs. relatives, the frequency of contact from various
Source Categories).  For repeated measures designs,
these data would be useful to compare sources of
support over time in terms of type and amount of support
available.

SUBSCALES AND VARIABLES

The scores entered for the variables listed in Table 1 are
combined to form subscales and variables.  Using the
Variable Names from Table 1, the combinations,
calculated by the computer, are described below.

Calculating Subscales

The original constructs of affect, affirmation, and aid are
each measured through the ratings made in response to
two questions.  As noted in the section on the 1995
revision, only two constructs emerged from factor
analysis.  Therefore, the items are combined into scores
for the subscales (Emotional and Tangible Support) as
follows:

EMO1 + EMO2 + EMO3 + EMO4 = EMOSUP
AID5 + AID6 = AID

Calculating Variables

Each of the three main variables is composed of 2-3
subscales.  The code name for Total Functional is
TLFUNCT; for Total Network Properties, TLNETWRK;
and for Total Loss, TLLOSS.  

Variable Subscales

TLFUNCT = EMOSUP + AID
TLNETWRK = NOLISTED + DURATION + FREQCON
TLLOSS = LOSS + LOSSNO + LOSSAMT

Missing Data

Occasionally a subject will not rate an individual network
member on a question.  This omission does not seriously
affect the subscale or variable in which the item is
imbedded.  However, if the subject does not give ratings
for an entire question, the subscale and/or variable
calculated from that question is not valid.  In this case,
only complete subscales or variables for that subject can
be used, and the unanswered question and its
corresponding subscale and/or variable cannot be used.
For example, if a subject did not answer Question 5, the
Aid subscale would be incomplete, and the variable, Total
Functional, cannot be calculated.  The other subscale



1995 NSSQ Scoring Instructions -- page 3

(Emotional Support) and the other two variables (Total
Network Properties and Total Loss) can still be used for
this subject.

SOURCE-SPECIFIC DATA

Additional information available from the NSSQ can be
analyzed to determine (a) the composition of the network
according to Source Category [SOU1 to SOU24], (b) the
categories of persons recently lost by the subject [LOSS1
to LOSS9], (c ) the amount of functional support provided
by each network member [PER1 to PER24], and (d) the
frequency of contact with each network member [CON1
to CON24].  The latter two types of information can be
calculated for the 9 Source Categories to determine the
amount of functional support provided by each category
and the frequency of contact with each category.

In some cases, the amount of variance explained by
source-specific categories is much greater than global
scores.  In a study of critical care nurses, for unmarried
nurses a specific source of support (from relatives)
explained 10% of the variance in perceived job stress and
16% of the variance in psychological symptoms--double
or triple the effect of the total functional support score
from the full network for the full sample (Norbeck, 1985).
Findings from a study of  low-income pregnant women
showed that specific sources of support (the women’s
mother or male partner) were predictive of pregnancy
outcomes for African-American women; whereas, the
total scores were not (Norbeck & Anderson, 1989).

SITUATION-SPECIFIC SUPPORT QUESTIONS

A final area of exploration in the use of the NSSQ has
been the incorporation of situation-specific questions for
particular populations.  In a study of work stress in critical
care nursing (Norbeck, 1985), two questions were
included on a half-page inserted after the Aid questions:
How much can you talk about your work with this person?
and How much does this person help you relax or re-
energize after work?  Findings from this study showed
that situation-specific support (work support) explained
24% of the variance of perceived job stress for the
married nurses--nearly double that of the total functional
support score for this group. 

Both the situation-specific and source-specific findings
are exploratory and need replication with other
populations.  The specific types or sources of support that
are most effective undoubtedly will vary by
sociodemographic characteristics of the population and
type of situation the population group is confronting.  If
similar results are obtained with different populations,
they will provide a basis for greater specificity in
assessing social support and in designing interventions
to augment low social support.

COMMENT ON THE USE OF AVERAGE SCORES

Some investigators have raised questions about the

impact of the format of the NSSQ which does not allow
separation of quantity of support from quality of support.
In this format, the more network members a person has,
the higher their support scores will be, regardless of the
ratings of the amount of support given for the individual
network members.

This concern, while valid from a theoretical perspective,
has not been born out by empirical experience;
furthermore, a demonstration of calculating average
scores shows that such scores create a bias against
individuals who report larger networks.  First, empirical
experience has shown that participants seldom name
individuals in their social support network who provide
low levels of support across the various types of support.
Thus, the more network members, the greater amount of
support -- both in terms of numbers of supporters and
quality of support.

Second, the fallacy of averaging (in an attempt to
separate quantity from quality) is that such scores result
in an artifact which biases against respondents who
report large numbers of supporters.  To illustrate, take
two respondents,  A and B.  Respondent A lists 4 network
members and gives each of these network members high
scores for quality of support.  Respondent B lists 14
network members and gives high scores for the first 8
network members, and  lower scores for the other
network members as shown:

A’s scores: 5 5 5 5
B’s scores: 5 5 5 5 5 5 5 5 4 4 3 3 2 2

The average score for Respondent A is  5, and the
average score for Respondent B is 4.14 -- yet
Respondent B had twice the number of network members
rated as providing the highest quality of support (5) and
other members who provided at least some support.

Although the SPSS program statements compute
average scores for use in methodological analyses, it is
theoretically and methodologically unwise to use them as
substantive variables.

Table 1

Computer Code Book for the NSSQ

Variable Name Field Width Variable Label Value Label

IDNO 4 Subject Number --
NOLISTED 2 Number Listed in Network --
EMO1 3 Emotional Support (Quest. 1) --
EMO2 3 Emotional Support (Quest. 2) --
EMO3 3 Emotional Support (Quest. 3) --
EMO4 3 Emotional Support (Quest. 4) --
AID5 3 Tangible Support (Quest. 5) --
AID6 3 Tangible Support (Quest. 6) --
DURATION 3 Duration of Relationships (Quest. 7) --
FREQCON 3 Frequency of Contact (Quest. 8) --

SOU1 1 Source Category Person 1
SOU2 1 Source Category Person 2
SOU3 1 Source Category Person 3
SOU4 1 Source Category Person 4
SOU5 1 Source Category Person 5
SOU6 1 Source Category Person 6 SOU1 to SOU24:
SOU7 1 Source Category Person 7
SOU8 1 Source Category Person 8 0 = none
SOU9 1 Source Category Person 9 1 = spouse or partner
SOU10 1 Source Category Person 10 2 = family or relatives
SOU11 1 Source Category Person 11 3 = friends
SOU12 1 Source Category Person 12 4 = work or school associates
SOU13 1 Source Category Person 13 5 = neighbors
SOU14 1 Source Category Person 14 6 = health care providers
SOU15 1 Source Category Person 15 7 = counselor or therapist
SOU16 1 Source Category Person 16 8 = minister/priest/rabbi
SOU17 1 Source Category Person 17 9 = other (e.g., God, pets)
SOU18 1 Source Category Person 18
SOU19 1 Source Category Person 19
SOU20 1 Source Category Person 20
SOU21 1 Source Category Person 21
SOU22 1 Source Category Person 22
SOU23 1 Source Category Person 23
SOU24 1 Source Category Person 24

LOSS 1 Recent Loss Question 9 0 = no, 1 = yes
LOSS1 2 Loss Spouse or Partner --
LOSS2 2 Loss Family or Relatives --
LOSS3 2 Loss Friends --
LOSS4 2 Loss Work or School Associates --
LOSS5 2 Loss Neighbors --
LOSS6 2 Loss Health Care Providers --
LOSS7 2 Loss Counselor or Therapist --
LOSS8 2 Loss Minister/Priest/Rabbi --
LOSS9 2 Loss Other --
LOSSNO 3 Number Lost --
LOSSAMT 1 Amount of Support Lost 0 = none

1 = a little
2 = moderate
3 = quite a bit
4 = a great deal

Table 2

Computer Code Book for Source-Specific Variables of the NSSQ

Variable Name Field Width Variable Label Value Label

PER1 2 Person 1 Total Functional --
PER2 2 Person 2 Total Functional --
PER3 2 Person 3 Total Functional --
PER4 2 Person 4 Total Functional --
PER5 2 Person 5 Total Functional --
PER6 2 Person 6 Total Functional --
PER7 2 Person 7 Total Functional --
PER8 2 Person 8 Total Functional --
PER9 2 Person 9 Total Functional --
PER10 2 Person 10 Total Functional --
PER11 2 Person 11 Total Functional --
PER12 2 Person 12 Total Functional --
PER13 2 Person 13 Total Functional --
PER14 2 Person 14 Total Functional --
PER15 2 Person 15 Total Functional --
PER16 2 Person 16 Total Functional --
PER17 2 Person 17 Total Functional --
PER18 2 Person 18 Total Functional --
PER19 2 Person 19 Total Functional --
PER20 2 Person 20 Total Functional --
PER21 2 Person 21 Total Functional --
PER22 2 Person 22 Total Functional --
PER23 2 Person 23 Total Functional --
PER24 2 Person 24 Total Functional --

CON1 1 Individual Contact Person 1
CON2 1 Individual Contact Person 2
CON3 1 Individual Contact Person 3
CON4 1 Individual Contact Person 4
CON5 1 Individual Contact Person 5
CON6 1 Individual Contact Person 6
CON7 1 Individual Contact Person 7
CON8 1 Individual Contact Person 8 CON1 to CON24:
CON9 1 Individual Contact Person 9
CON10 1 Individual Contact Person 10 5 = daily
CON11 1 Individual Contact Person 11 4 = weekly
CON12 1 Individual Contact Person 12 3 = monthly
CON13 1 Individual Contact Person 13 2 = a few times a year
CON14 1 Individual Contact Person 14 1 = once a year or less
CON15 1 Individual Contact Person 15
CON16 1 Individual Contact Person 16
CON17 1 Individual Contact Person 17
CON18 1 Individual Contact Person 18
CON19 1 Individual Contact Person 19
CON20 1 Individual Contact Person 20
CON21 1 Individual Contact Person 21
CON22 1 Individual Contact Person 22
CON23 1 Individual Contact Person 23
CON24 1 Individual Contact Person 24

        Figure 1
Subject No.                        [IDNO] SOCIAL SUPPORT QUESTIONNAIRE              Demographic Data:  Age        Sex         Ethnicity          
No. in Network                        [NOLISTED]        Supplemental Scoring Sheet              Education (yrs)                   Marital Status                   

1995 Version Type of Group                           (e.g. dialysis patients) 

Relationship Question 1 Question 2 Question 3 Question 4 Question 5 Question  6 Person Totals (Ques. 1-6) Question 7 Questions 8

1 [SOU1] [PER1] [CON1]

2 [SOU2] [PER2] [CON2]

3 [SOU3] [PER3] [CON3]

4 [SOU4] [PER4] [CON4]

5 [SOU5] [PER5] [CON5]

6 [SOU6] [PER6] [CON6]

7 [SOU7] [PER7] [CON7]

8 [SOU8] [PER8] [CON8]

9 [SOU9] [PER9] [CON9]

10 [SOU10] [PER10] [CON10]

11 [SOU11] [PER11] [CON11]

12 [SOU12] [PER12] [CON12]

13 [SOU13] [PER13] [CON13]

14 [SOU14] [PER14] [CON14]

15 [SOU15] [PER15] [CON15]

16 [SOU16] [PER16] [CON16]

17 [SOU17] [PER17] [CON17]

18 [SOU18] [PER18] [CON18]

19 [SOU19] [PER19] [CON19]

20 [SOU20] [PER20] [CON20]

21 [SOU21] [PER21] [CON21]

22 [SOU22] [PER22] [CON22]

23 [SOU23] [PER23] [CON23]

24 [SOU24] [PER24] [CON24]

Question Totals
      [EMO1] [EMO2]        [EMO3] [EMO4]           [AID5]       [AID6]                       [DURATION]            [FREQCON]

[LOSS]   QUESTION 9    _________ 0 = No, 1 = Yes
[LOSSNO]   QUESTION 9a  _________ Category codes: [LOSS1] _____ [LOSS2] ______ [LOSS3] _____ [LOSS4] _____ [LOSS5] _____ [LOSS6] _____ [LOSS7] _____ [LOSS8] ______
[LOSS9] ______
[LOSSAMT]  QUESTION 9b _________

Appendix A

Additional Analyses with the NSSQ

1.  Corrected Normative Scores for Previously Published Data

The previously published normative data for 136 employed adults (Norbeck, Lindsey, & Carrieri,
1983) were based on the uncorrected scale values for Questions 1-6.  The Scoring Adjustment
(see page 1 of this document) applied to the data published in Table 2 in the 1983 article resulted
in the following corrected (lower) mean values:

Table A-1.   Corrected Scores for Normative Data Published in 1983

  Mean S.D.
NSSQ Scales:

Affect   73.49 36.25
Affirmation   66.06 32.33
   (Emotional Support 139.55)
Aid   62.35 32.24

Total Functional  201.90 95.87

As described in the scoring instructions, the 1984 Scoring Adjustment applied only to questions
1-6 (Affect, Affirmation, and Aid); thus, the mean values for the other NSSQ subscales and
variables on Table 2 in the 1983 article were not changed.  Scores obtained from the 1995
version of the NSSQ do not need this correction because the 5-point rating scale for Questions
1-6 was changed to a 0-4 scale on the instrument.

2.  Factor Structure of the NSSQ

Dr. Suzanne Dibble and Dr. Steven Paul collaborated in examining the factor structure of the
NSSQ.  Data shared by investigators who used the NSSQ were pooled resulting in a total sample
of 1,392 participants with complete data.  A principal axes factor analysis with varimax rotation
was conducted.  This factor analytic procedure identified two distinct dimensions of social
support:  emotional support and tangible support.  This two factor model accounts for 74% of the
total variance.  The loadings of the 6 items on the two rotated factors are reported in Table A-2.
Factor loadings exceeding 0.40 are highlighted.

Table A-2.  Factor Pattern Following Principal Axes Factoring with Varimax Rotation

Average          Factor 1        Factor 2
Scores Emotional Support Tangible Support

Question 2 .83 .16
Question 1 .78 .20
Question 4 .68 .31
Question 3 .64 .35

Question 5 .21 .79
Question 6 .24 .70 

Appendix A (cont.)

3.  Normative Values for the NSSQ (Healthy Adults)
  

Dr. Suzanne Dibble examined the Normative Values of the NSSQ.  Data from healthy participants
shared by investigators who used the NSSQ were pooled resulting in a total sample of 1,415
participants.  Data were not available for the number of persons lost (Number Lost) in the
previous year for most of these participants; therefore, the Total Loss Score could not be
calculated.

Table A-3.  Normative Sample Demographics

Females Males
(n=1,067) (n=348) 

Demographic Variables: Mean S.D. . . . . . . . Mean S.D.  

Age 45.5 21.0 . . . . . . . 48.3 17.3
Education Level 14.0 3.2 . . . . . . . . 14.9   3.9

Percent Caucasian 78.6% . . . . . . . . . . 88.8%
Percent Married/Partnered 40.5% . . . . . . . . . . 53.2%   

Table A-4.  Normative Values for the NSSQ (Healthy Adults)

Females . . . . . . . . . . Males
(n=1,067) . . . . . . . . . (n=348)   

NSSQ Variables: Mean S.D. . . . . . . . Mean S.D.

Number Listed 10.9 . . . 5.9 10.6 . . 6.0
Emotional Support 127.2 . . 72.7 119.3 . 75.2
Tangible Support  53.1 . . 33.4 55.3 . 36.0
Total Functional Support    179.4 . 102.1 173.6 108.3
Total Network Score 98.5 . . 53.8 95.0 . 55.3
Amount of Loss 2.4 . . . 1.3 2.2 . . 1.1

Percent Experienced Loss 44.1% . . . 36.3%

Appendix B

Scoring Adjustment for the 1980/82 Version of the NSSQ

Background These SPSS program statements can be downloaded:  NSSQafix.sps

Following publication of the instrument and scoring instructions in 1980/82, a Scoring Adjustment
was made in 1984 for scoring Questions 1-6.  For these questions, the 5-point rating scale was
converted in the computer to values of 0-4 scale, rather than the values of 1-5 printed on the
instrument.  This conversion was necessary because the rating of "1" equals "not at all" (no
support) from a network member.  However, the "1s" would continue to be added to the score,
thereby artificially inflating the total amount of support.  Revised means and standard deviations
for the normative data published in 1983 are shown in Appendix A.  The 1995 revised version of
the NSSQ eliminated the need for the Scoring Adjustment by changing the rating scale from 1-5
to 0-4 directly on the instrument.

The scoring conversion provided below corrects the scores for Questions 1-6 and includes a
parallel scoring conversion required for the source-specific scores derived from the functional
support items.  Scoring adjustment was not necessary for questions 7 and 8 because the rating
of "1" has a non-zero value for those questions. 

Using the 1995 Windows SPSS scoring statements with earlier versions of the NSSQ

For data collected with the 1980/82 version of the NSSQ, the following SPSS program statements must
be applied to the data before using the 1995 scoring instructions provided in Appendix C.  Prior to
executing the scoring conversion, the variable names for the Emotional Support items are changed to
correspond to the revised format of the 1995 instrument.

COMMENT
Syntax Filename = NSSQFIXA.SPS
The following SPSS 6.1 for Windows syntax statements carry out the
variable renaming and scoring adjustments necessary to make the data file
compatible with the 1995 revised instrument and SPSS scoring statements in
Appendix C.

TITLE 'SCORING ADJUSTMENT FOR THE 1980/82 VERSION OF THE NSSQ'.
SUBTITLE 'INCLUDING VARIABLE RENAMING STATEMENTS'.

COMMENT
Provide the complete path to the SPSS for Windows data file that contains the
variables of the 1980/82 version of the NSSQ in the GET FILE command.  If the
data file has already been opened in SPSS, it is not necessary to GET the file
again.

GET FILE = c:\subdirectory\filename.sav'.

COMMENT
The following commands rename variables to correspond with the 1995 version
of the Scoring Instructions.

RENAME VARIABLES (AFFECT1 AFFECT2 AFFIRM3 AFFIRM4 = EMO1 EMO2 EMO3 EMO4)
                 (SOURCE1 TO SOURCE24 = SOU1 TO SOU24)
                 (PERSON1 TO PERSON24 = PER1 TO PER24)
                 (CONTAC1 TO CONTAC24 = CON1 TO CON24).

Appendix B (cont.)

COMMENT
The following commands create the scoring adjustment for EMO1 to AID6 by
subtracting the number of sources listed in the network.  These adjustments
are necessary if the 1980/82 version of the NSSQ was used.

COMPUTE EMO1=EMO1-NOLISTED /* CORRECTED EMO1 SCORE */.
COMPUTE EMO2=EMO2-NOLISTED /* CORRECTED EMO2 SCORE */.
COMPUTE EMO3=EMO3-NOLISTED /* CORRECTED EMO3 SCORE */.
COMPUTE EMO4=EMO4-NOLISTED /* CORRECTED EMO4 SCORE */.
COMPUTE AID5=AID5-NOLISTED /* CORRECTED AID5 SCORE */.
COMPUTE AID6=AID6-NOLISTED /* CORRECTED AID6 SCORE */.

COMMENT
The following commands create the scoring adjustment for PER1 to PER24 by
subtracting 6 from each score.

DO REPEAT XPER=PER1 TO PER24.
COMPUTE XPER=XPER-6 /* CORRECTED PERSON SCORES */.
END REPEAT.
EXECUTE.

COMMENT
Provide a NEW filename for the SPSS data file that will contain the renamed
and adjusted variables with the SAVE OUTFILE command.

SAVE OUTFILE = 'c:\subdirectory\newname.sav'/COMPRESSED.

Appendix C

SPSS Command Statements to Create Subscales, Variables, and Source-Specific Variables

The computer statements that follow were based on program statements initially developed for the
mainframe SPSS by Dr. Don Chambers and revised in 1995 for the Windows version of SPSS by Dr.
Steven Paul.  The purpose of each of the functions in these sets of command statements is briefly
explained in the “Comment” preceding it.  

Note: If these revised 1995 program statements are to be used with the 1980/82 version of the NSSQ,
the commands in Appendix B must be executed first.

These SPSS program statements can be downloaded:  NSSQ95a.sps

1.  Establishing Variables from Raw Data

COMMENT
 Syntax Filename  = NSSQ95A.SPS

The following SPSS 6.1 for Windows syntax statements assigns variable
and value labels, computes various subscale and total scores,
checks for out-of-range values, provides basic descriptive statistics,
and lists subjects with discrepant data. 

TITLE "SCORING PROGRAM FOR 'NORBECK SOCIAL SUPPORT QUESTIONNAIRE'" .
SUBTITLE 'INCLUDING CHECK FOR DISCREPANT DATA' .

COMMENT 
Provide the complete path to the SPSS for Windows data file that
contains the variables of the 1995 version of the NSSQ in the 
GET FILE command.  If the data file has already been opened in SPSS,

 it is not necessary to GET the file again.

GET FILE = 'c:\subdirectory\filename.sav' .

VARIABLE  LABELS 
 IDNO 'SUBJECT  NUMBER'       
 NOLISTED 'NUMBER LISTED IN NETWORK'  
 EMO1 'EMOTIONAL SUPPORT (QUEST. 1)' 
 EMO2 'EMOTIONAL SUPPORT (QUEST. 2)' 
 EMO3 'EMOTIONAL SUPPORT (QUEST. 3)' 
 EMO4 'EMOTIONAL SUPPORT (QUEST. 4)' 
 AID5 'TANGIBLE SUPPORT (QUEST. 5)' 
 AID6 'TANGIBLE SUPPORT (QUEST. 6)' 
 DURATION 'DURATION OF RELATIONSHIP (QUEST. 7)' 
 FREQCON 'FREQUENCY OF CONTACT (QUEST. 8)' 
 LOSS 'RECENT LOSS (QUEST. 9)' 
 LOSS1 'NUMBER OF SPOUSES OR PARTNERS LOST' 
 LOSS2 'NUMBER OF FAMILY OR RELATIVES LOST' 
 LOSS3 'NUMBER OF FRIENDS LOST' 
 LOSS4 'NUMBER OF WORK OR SCHOOL ASSOCIATES LOST' 
 LOSS5 'NUMBER OF NEIGHBORS LOST' 
 LOSS6 'NUMBER OF HEALTH CARE PROVIDERS LOST' 
 LOSS7 'NUMBER OF COUNSELORS OR THERAPISTS LOST' 
 LOSS8 'NUMBER OF MINISTERS/PRIESTS/RABBIS LOST' 
 LOSS9 'NUMBER OF OTHERS LOST'  
 LOSSNO  'TOTAL NUMBER OF PERSONS LOST' 
 LOSSAMT  'AMOUNT OF SUPPORT LOST (QUEST. 9b)'.

VALUE  LABELS 
 SOU1 TO SOU24 0 'NONE' 1 'SPOUSE OR PARTNER' 2 'FAMILY OR RELATIVES'

3 'FRIENDS' 4 'WORK/SCHOOL ASSOC' 5 'NEIGHBORS' 
6 'HLTH CARE PROVIDERS' 7 'COUNSELOR/THERAPIST'
8 'MINIST/PRIEST/RABBI' 9 'OTHER'/ 

 LOSS 0 'NO' 1 'YES'/ 

LOSSAMT 0 'NONE AT ALL' 1 'A LITTLE'  2 'A MODERATE AMOUNT' 
3 'QUITE A BIT' 4 'A GREAT  DEAL'/

 CON1 TO CON24 5 'DAILY' 4 'WEEKLY' 3 'MONTHLY' 2 'A FEW TIMES A YEAR' 
 1 'ONCE A YEAR OR LESS'.

COMMENT 
The following commands calculate subscale and total scores. 

COMPUTE EMOSUP=EMO1+EMO2+EMO3+EMO4 /* EMOTIONAL SUPPORT SCORE */.
COMPUTE AID=AID5+AID6 /* TANGIBLE SUPPORT SCORE */.
COMPUTE TLFUNCT=EMOSUP+AID /* TOTAL FUNCTIONAL SUPPORT */.
COMPUTE TLNETWRK=NOLISTED+DURATION+FREQCON /* TOTAL NETWORK SCORE
*/.
COMPUTE TLLOSS=LOSS+LOSSNO+LOSSAMT /* TOTAL LOSS SCORE */.

COMMENT
The following two variables calculate Average Duration and Average

 Functional Support by dividing by the number of sources listed in the 
 network.

COMPUTE AVEDURA=DURATION/NOLISTED /* AVERAGE DURATION SCORE */.
COMPUTE AVEFUNCT=TLFUNCT/NOLISTED /* AVE. FUNCT. SUPPORT
SCORE */.

COMMENT 
TLFUNCT2 is a check on TLFUNCT -- ideally, the two scores should be

 identical because they are simply two different ways of calculating
 the same variable. 

COMPUTE TLFUNCT2=SUM(PER1 TO PER24) /* CHECK SCORE FOR TLFUNCT */.

COMMENT 
The following commands check for out-of-range values.  If any subject has
out-of-range values, they are coded "1" for the variable "PROBLEM".

IF (NOLISTED <0 OR NOLISTED >24) PROBLEM=1.

DO REPEAT XSCORE=EMO1 TO AID6.
IF (XSCORE <0 OR XSCORE > NOLISTED*4) PROBLEM=2.
END REPEAT.

IF (DURATION < NOLISTED OR DURATION > NOLISTED*5) PROBLEM=3.
IF (FREQCON < NOLISTED OR FREQCON > NOLISTED*5) PROBLEM=4.

DO REPEAT XSOU=SOU1 TO SOU24/ 
          XPER=PER1 TO PER24/
          XCON=CON1 TO CON24. 
IF (XSOU <0 OR XSOU >9) PROBLEM=5.
IF (XPER <0 OR XPER >24) PROBLEM=6.
IF (XCON <1 OR XCON >5) PROBLEM=7.
END REPEAT.

VARIABLE  LABELS 
 EMOSUP 'EMOTIONAL SUPPORT SCORE'
 AID 'TANGIBLE SUPPORT SCORE' 
 TLFUNCT 'TOTAL FUNCTIONAL  SUPPORT SCORE' 
 TLNETWRK 'TOTAL NETWORK  SCORE' 
 TLLOSS 'TOTAL LOSS  SCORE'
 AVEDURA 'AVERAGE DURATION  SCORE'
 AVEFUNCT 'AVERAGE FUNCTIONAL SUPPORT  SCORE'
 TLFUNCT2 'CHECK SCORE FOR  TLFUNCT' 
 PROBLEM 'CHECK VARIABLE FOR OUT-OF-RANGE VALUES'.

VALUE  LABELS 
 PROBLEM 1 'NOLISTED OUT-OF-RANGE' 2 'EMO1 to AID6 OUT-OF-RANGE'
         3 'DURATION OUT-OF RANGE' 4 'FREQCON OUT-OF-RANGE'
         5 'SOU1 TO SOU24 OUT-0F-RANGE'
         6 'PER1 TO PER24 OUT-OF-RANGE'
         7 'CON1 TO CON24 OUT-OF-RANGE'.

COMMENT
Provide a NEW filename for the SPSS data file that will contain the 
original and newly created variables with the SAVE OUTFILE command.

SAVE OUTFILE = 'c:\subdirectory\filesub1.sav' /COMPRESSED.

COMMENT 
The following two commands provide basic descriptive statistics. 

FREQUENCIES VAR=PROBLEM NOLISTED LOSS LOSSNO LOSSAMT.

DESCRIPTIVES VAR=NOLISTED TO FREQCON LOSS1 TO LOSSAMT EMOSUP TO TLFUNCT2.

COMMENT 
 The last set of commands lists subjects with discrepant data. 

FILTER BY PROBLEM.
LIST VAR=IDNO  NOLISTED TO  FREQCON.
LIST VAR=IDNO  SOU1 TO  SOU24.
LIST VAR=IDNO  PER1 TO  PER24.
LIST VAR=IDNO  CON1 TO  CON24.
FILTER OFF.

2.  Calculation of Source-Specific Scores

These SPSS program statements can be downloaded:  NSSQ95b.sps

The variables calculated from this set of statements allow comparison of the amount of Total Functional
Support (the sum of Emotional Support and Tangible Support) that is available from friends, relatives, or
other source categories.  Similar comparisons can be made for Frequency of Contact with various source
categories of network members.  These calculations generate a large number of variables, some of which
are intermediate variables that can be erased after the associated summary score is obtained.

COMMENT
Syntax Filename  = NSSQ95B.SPS
The following SPSS 6.1 for Windows syntax statements creates 
source-specific functional support and contact scores.

TITLE 'CALCULATION OF VARIOUS TYPES OF FUNCTIONAL SUPPORT AND'.
SUBTITLE 'CONTACT SCORES BY SOURCE CATEGORY'.

COMMENT
Provide the complete path to the SPSS for Windows data file that was
created by the syntax commands in NSSQ95A.SPS.  If the data file has already
been opened in SPSS, it is not necessary to GET the file again.

GET FILE = 'c:\subdirectory\filesub1.sav' .
 
COMMENT 

The following commands create 9 new variables which indicate Total
Functional Support broken down by each of the 9 source categories.

DO REPEAT XTOTPFOR=TOTPFOR1 TO TOTPFOR9.
COMPUTE XTOTPFOR=0.
END REPEAT.

DO REPEAT XSOU=SOU1 TO SOU24/XPER=PER1 TO PER24.
IF (XSOU EQ 1) TOTPFOR1=TOTPFOR1+XPER.
IF (XSOU EQ 2) TOTPFOR2=TOTPFOR2+XPER.
IF (XSOU EQ 3) TOTPFOR3=TOTPFOR3+XPER. 
IF (XSOU EQ 4) TOTPFOR4=TOTPFOR4+XPER.
IF (XSOU EQ 5) TOTPFOR5=TOTPFOR5+XPER. 
IF (XSOU EQ 6) TOTPFOR6=TOTPFOR6+XPER. 
IF (XSOU EQ 7) TOTPFOR7=TOTPFOR7+XPER. 
IF (XSOU EQ 8) TOTPFOR8=TOTPFOR8+XPER. 
IF (XSOU EQ 9) TOTPFOR9=TOTPFOR9+XPER. 
END REPEAT.

COMMENT 
The following commands create 9 new variables which indicate Total Contact
broken down by each of the 9 source categories.  

 
DO REPEAT XTOTCFOR=TOTCFOR1 TO TOTCFOR9.
COMPUTE XTOTCFOR=0.
END REPEAT.

DO REPEAT XSOU=SOU1 TO SOU24/XCON=CON1 TO CON24.
IF (XSOU EQ 1) TOTCFOR1=TOTCFOR1+XCON. 

IF (XSOU EQ 2) TOTCFOR2=TOTCFOR2+XCON. 
IF (XSOU EQ 3) TOTCFOR3=TOTCFOR3+XCON. 
IF (XSOU EQ 4) TOTCFOR4=TOTCFOR4+XCON. 
IF (XSOU EQ 5) TOTCFOR5=TOTCFOR5+XCON. 
IF (XSOU EQ 6) TOTCFOR6=TOTCFOR6+XCON. 
IF (XSOU EQ 7) TOTCFOR7=TOTCFOR7+XCON. 
IF (XSOU EQ 8) TOTCFOR8=TOTCFOR8+XCON. 
IF (XSOU EQ 9) TOTCFOR9=TOTCFOR9+XCON. 
END REPEAT.

COMMENT 
The following commands create 9 new variables which count the number of
persons listed in the network for each of the 9 source categories. 

 
DO REPEAT XCOUNT=COUNT1 TO COUNT9/X=1 TO 9.
COUNT XCOUNT=SOU1 TO SOU24(X).
END REPEAT.

COMMENT 
The following commands create 9 new variables which indicate Average
Functional Support for each of the 9 source categories. 

 
DO REPEAT XAVEPFOR=AVEPFOR1 TO AVEPFOR9/ 
          XTOTPFOR=TOTPFOR1 TO TOTPFOR9/ 
          XCOUNT=COUNT1 TO COUNT9.
IF (XCOUNT NE 0) XAVEPFOR=XTOTPFOR/XCOUNT.
END REPEAT.

COMMENT 
The following commands create 9 new variables which indicate Average
Contact for each of the 9 source categories. 

DO REPEAT XAVECFOR=AVECFOR1 TO AVECFOR9/ 
          XTOTCFOR=TOTCFOR1 TO TOTCFOR9/ 
          XCOUNT=COUNT1 TO COUNT9.
IF (XCOUNT NE 0) XAVECFOR=XTOTCFOR/XCOUNT.
END REPEAT.

COMMENT 
The following commands create 9 new variables which indicate the percent
of the total number of persons listed in the network broken down by each
of the 9 source categories. 

DO REPEAT XPERCNOL=PERCNOL1 TO PERCNOL9/XCOUNT=COUNT1 TO COUNT9.
COMPUTE XPERCNOL=(XCOUNT/NOLISTED)*100.
END REPEAT.

COMMENT 
The following commands create 9 new variables which indicate the percent
of the Total Functional Support in the network broken down by each of the
9 source categories. 

DO REPEAT XPERCFUN=PERCFUN1 TO PERCFUN9/XTOTPFOR=TOTPFOR1 TO TOTPFOR9.
COMPUTE XPERCFUN=(XTOTPFOR/TLFUNCT)*100.
END REPEAT.



1995 NSSQ Scoring Instructions -- page 16

Appendix C (cont.)

VARIABLE LABELS 
 TOTPFOR1 'TOT FUNCT SUPP FOR SPOUSE OR PARTNER' 
 TOTPFOR2 'TOT FUNCT SUPP FOR FAMILY OR RELATIVES' 
 TOTPFOR3 'TOT FUNCT SUPP FOR FRIENDS' 
 TOTPFOR4 'TOT FUNCT SUPP FOR WORK/SCHOOL ASSOC' 
 TOTPFOR5 'TOT FUNCT SUPP FOR NEIGHBORS' 
 TOTPFOR6 'TOT FUNCT SUPP FOR HLTH CARE PROV' 
 TOTPFOR7 'TOT FUNCT SUPP FOR COUNSELOR OR THERAP' 
 TOTPFOR8 'TOT FUNCT SUPP FOR MINIST/PRIEST/RABBI' 
 TOTPFOR9 'TOT FUNCT SUPP FOR OTHER' 
 TOTCFOR1 'TOT CONTACT FOR SPOUSE OR PARTNER' 
 TOTCFOR2 'TOT CONTACT FOR FAMILY OR RELATIVES' 
 TOTCFOR3 'TOT CONTACT FOR FRIENDS' 
 TOTCFOR4 'TOT CONTACT FOR WORK/SCHOOL ASSOC' 
 TOTCFOR5 'TOT CONTACT FOR NEIGHBORS' 
 TOTCFOR6 'TOT CONTACT FOR HLTH CARE PROV' 
 TOTCFOR7 'TOT CONTACT FOR COUNSELOR OR THERAP' 
 TOTCFOR8 'TOT CONTACT FOR MINIST/PRIEST/RABBI' 
 TOTCFOR9 'TOT CONTACT FOR OTHER' 
 COUNT1 'NUMBER IN NETWORK -- SPOUSE OR PARTNER' 
 COUNT2 'NUMBER IN NETWORK -- FAMILY OR RELATIVES' 
 COUNT3 'NUMBER IN NETWORK -- FRIENDS' 
 COUNT4 'NUMBER IN NETWORK -- WORK/SCHOOL ASSOC' 
 COUNT5 'NUMBER IN NETWORK -- NEIGHBORS' 
 COUNT6 'NUMBER IN NETWORK -- HLTH CARE PROV' 
 COUNT7 'NUMBER IN NETWORK -- COUNSELOR OR THERAP' 
 COUNT8 'NUMBER IN NETWORK -- MINIST/PRIEST/RABBI' 
 COUNT9 'NUMBER IN NETWORK -- OTHER' 
 AVEPFOR1 'AVE FUNCT SUPP FOR SPOUSE OR PARTNER' 
 AVEPFOR2 'AVE FUNCT SUPP FOR FAMILY OR RELATIVES' 
 AVEPFOR3 'AVE FUNCT SUPP FOR FRIENDS' 
 AVEPFOR4 'AVE FUNCT SUPP FOR WORK/SCHOOL ASSOC' 
 AVEPFOR5 'AVE FUNCT SUPP FOR NEIGHBORS' 
 AVEPFOR6 'AVE FUNCT SUPP FOR HLTH CARE PROV' 
 AVEPFOR7 'AVE FUNCT SUPP FOR COUNSELOR OR THERAP' 
 AVEPFOR8 'AVE FUNCT SUPP FOR MINIST/PRIEST/RABBI' 
 AVEPFOR9 'AVE FUNCT SUPP FOR OTHER' 
 AVECFOR1 'AVE CONTACT FOR SPOUSE OR PARTNER' 
 AVECFOR2 'AVE CONTACT FOR FAMILY OR RELATIVES' 
 AVECFOR3 'AVE CONTACT FOR FRIENDS' 
 AVECFOR4 'AVE CONTACT FOR WORK/SCHOOL ASSOC' 
 AVECFOR5 'AVE CONTACT FOR NEIGHBORS' 
 AVECFOR6 'AVE CONTACT FOR HLTH CARE PROV' 
 AVECFOR7 'AVE CONTACT FOR COUNSELOR OR THERAP' 
 AVECFOR8 'AVE CONTACT FOR MINIST/PRIEST/RABBI' 
 AVECFOR9 'AVE CONTACT FOR OTHER' 
 PERCNOL1 '% OF NOLISTED FOR SPOUSE OR PARTNER' 
 PERCNOL2 '% OF NOLISTED FOR FAMILY OR RELATIVES' 
 PERCNOL3 '% OF NOLISTED FOR FRIENDS' 
 PERCNOL4 '% OF NOLISTED FOR WORK/SCHOOL ASSOC' 
 PERCNOL5 '% OF NOLISTED FOR NEIGHBORS' 
 PERCNOL6 '% OF NOLISTED FOR HLTH CARE PROV' 
 PERCNOL7 '% OF NOLISTED FOR COUNSELOR OR THERAP' 
 PERCNOL8 '% OF NOLISTED FOR MINIST/PRIEST/RABBI' 
 PERCNOL9 '% OF NOLISTED FOR OTHER' 
 PERCFUN1 '% OF TLFUNCT FOR SPOUSE OR PARTNER' 

 PERCFUN2 '% OF TLFUNCT FOR FAMILY OR RELATIVES' 
 PERCFUN3 '% OF TLFUNCT FOR FRIENDS' 
 PERCFUN4 '% OF TLFUNCT FOR WORK/SCHOOL ASSOC' 
 PERCFUN5 '% OF TLFUNCT FOR NEIGHBORS' 
 PERCFUN6 '% OF TLFUNCT FOR HLTH CARE PROV' 
 PERCFUN7 '% OF TLFUNCT FOR COUNSELOR OR THERAP' 
 PERCFUN8 '% OF TLFUNCT FOR MINIST/PRIEST/RABBI' 
 PERCFUN9 '% OF TLFUNCT FOR OTHER' 

COMMENT
Provide a NEW filename for the SPSS data file that will contain the 
original and newly created variables with the SAVE OUTFILE command.

SAVE OUTFILE = 'c:\subdirectory\filesub2.sav' /COMPRESSED.

COMMENT 
The following two commands provide basic descriptive statistics. 

FREQUENCIES VAR=COUNT1 TO COUNT9.

DESCRIPTIVES VAR=TOTPFOR1 TO PERCFUN9.
 


